
LMS
GALAHAD Optimization Library version 3.0

SUMMARY

Given a sequence of vectors {sk } and {yk } and scale factors δk , obtain the product of a limited-memory secant
approximation Hk (or its inverse) with a given vector, using one of a variety of well-established formulae.
ATTRIBUTES — Versions: GALAHAD LMS single, GALAHAD LMS double. Uses: GALAHAD CLOCK, GALAHAD SYMBOLS, GALAHAD SPACE, GALAHAD LAPACK interface, GALAHAD BLAS interface, GALAHAD SPECFILE. Date: July
2014. Origin: N. I. M. Gould, Rutherford Appleton Laboratory. Language: Fortran 95 + TR 15581 or Fortran 2003.

HOW TO USE THE PACKAGE

Access to the package requires a USE statement such as
Single precision version
USE GALAHAD LMS single
Double precision version
USE GALAHAD LMS double
If it is required to use both modules at the same time, the derived types SMT type, LMS time type, LMS control type,
LMS inform type and LMS data type (§2.1) and the subroutines LMS initialize, LMS setup, LMS form, LMS form shift, LMS apply, LMS terminate, (§2.2) and LMS read specfile (§2.4) must be renamed on one of the USE
statements.
2.1 The derived data types
Four derived data types are accessible from the package.
2.1.1 The derived data type for holding control parameters
The derived data type LMS control type is used to hold controlling data. Default values may be obtained by calling LMS initialize (see §2.2.1), while components may also be changed by calling GALAHAD LMS read spec (see
§2.4.1). The components of LMS control type are:
error is a scalar variable of type default INTEGER, that holds the stream number for error messages. Printing of error
messages in LMS setup, LMS apply and LMS terminate is suppressed if error ≤ 0. The default is error =
6.
out is a scalar variable of type default INTEGER, that holds the stream number for informational messages. Printing
of informational messages in LMS setup, LMS apply is suppressed if out < 0. The default is out = 6.
print level is a scalar variable of type default INTEGER, that is used to control the amount of informational output
which is required. No informational output will occur if print level ≤ 0. If print level = 1, a single line
of output will be produced for each level of the process. If print level ≥ 2, this output will be increased to
provide significant detail of the factorization. The default is print level = 0.

All use is subject to licence. See http://galahad.rl.ac.uk/galahad-www/cou.html .
For any commercial application, a separate license must be signed.
GALAHAD

LMS (February 20, 2017)

1

LMS

GALAHAD

memory length is a scalar variable of type default INTEGER, that is used to specify the maximum number of vectors {sk } and {yk } that will be used when building the secant approximation. Any non-positive value will be
interpreted as 1. The default is memory length = 10.
method is a scalar variable of type default INTEGER, that is used to specify the limited-memory formula that will be
applied. Possible values are
1 . A limited-memory BFGS formula will be applied.
2 . A limited-memory symmetric rank-one formula will be applied.
3 . The inverse of the limited-memory BFGS formula will be applied.
4 . The inverse of the shifted limited-memory BFGS formula will be applied. This should be used instead of
%method = 3 whenever a shift is planned.
Any value outside this range will be interpreted as 1. The default is method = 1.
any method is a scalar variable of type default LOGICAL, that must be set .TRUE. if more than one method (see
%method above) is to be used and .FALSE. otherwise. The package will require more storage and may run
slower if any method is .TRUE.. The default is any method = .FALSE..
space critical is a scalar variable of type default LOGICAL, that must be set .TRUE. if space is critical when
allocating arrays and .FALSE. otherwise. The package may run faster if space critical is .FALSE. but at the
possible expense of a larger storage requirement. The default is space critical = .FALSE..
deallocate error fatal is a scalar variable of type default LOGICAL, that must be set .TRUE. if the user wishes to
terminate execution if a deallocation fails, and .FALSE. if an attempt to continue will be made. The default is
deallocate error fatal = .FALSE..
prefix is a scalar variable of type default CHARACTER and length 30, that may be used to provide a user-selected
character string to preface every line of printed output. Specifically, each line of output will be prefaced by the
string prefix(2:LEN(TRIM(prefix))-1), thus ignoring the first and last non-null components of the supplied
string. If the user does not want to preface lines by such a string, they may use the default prefix = "".
2.1.2 The derived data type for holding timing information
The derived data type LMS time type is used to hold elapsed CPU and system clock times for the various parts of the
calculation. The components of LMS time type are:
total is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the total CPU
time spent in the package.
setup is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the CPU time
spent setting up the data structures to represent the limited-memory matrix.
form is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the CPU time
spent forming and updating the limited-memory matrix as new data arrives.
apply is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the CPU time
spent applying the matrix to given vectors.
clock total is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the total
elapsed system clock time spent in the package.
clock setup is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the
elapsed system clock time spent setting up the data structures to represent the limited-memory matrix.

clock form is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the
elapsed system clock time spent forming and updating the limited-memory matrix as new data arrives.
clock apply is a scalar variable of type default REAL (double precision in GALAHAD LMS double), that gives the
elapsed system clock time spent applying the matrix to given vectors. factor R.
2.1.3 The derived data type for holding informational parameters
The derived data type LMS inform type is used to hold parameters that give information about the progress and needs
of the algorithm. The components of LMS inform type are:
status is a scalar variable of type default INTEGER, that gives the exit status of the algorithm. See §2.3 for details.
alloc status is a scalar variable of type default INTEGER, that gives the status of the last attempted array allocation
or deallocation. This will be 0 if status = 0.
bad alloc is a scalar variable of type default CHARACTER and length 80, that gives the name of the last internal array
for which there were allocation or deallocation errors. This will be the null string if status = 0.
length is a scalar variable of type default INTEGER, that gives the number of pairs {sk , yk } currently used to represent
the limited-memory matrix
updates skipped is a scalar variable of type default LOGICAL, that will be .TRUE. if one or more of the current pairs
{sk , yk } has been ignored for stability reasons when building the current limited-memory matrix, and .FALSE.
otherwise.
time is a scalar variable of type LMS time type whose components are used to hold elapsed CPU and system clock
times for the various parts of the calculation (see Section 2.1.2).
2.1.4 The derived data type for holding problem data
The derived data type LMS data type is used to hold all the data for the problem and the workspace arrays used to
construct the multi-level incomplete factorization between calls of LMS procedures. This data should be preserved,
untouched, from the initial call to LMS initialize to the final call to LMS terminate.
2.2 Argument lists and calling sequences
There are seven procedures for user calls (see §2.4 for further features):
1. The subroutine LMS initialize is used to set default values, and initialize private data, before solving one or
more problems with the same sparsity and bound structure.
2. The subroutine LMS setup is called to set up the data structures needed to represent the limited-memory matrix
Hk or its inverse.
3. The subroutine LMS form is called to form the limited-memory matrix Hk or its inverse as new data (sk , yk , δk )
arrives. The matrix Hk + λk I or its inverse for a specified shift λk may be formed instead.
4. The subroutine LMS form shift is called to update the inverse of the limited-memory matrix Hk + λk I when a
new shift λk is required.
5. The subroutine LMS change method is called to build the limited-memory matrices Hk , Hk + λk I or their inverse
for a new method from the current data.

6. The subroutine LMS apply is called to form the product u = Hk v, u = (Hk + λk I)v, u = H−1
k v or u = (Hk +
λk I)−1 v for a given vector v.
7. The subroutine LMS terminate is provided to allow the user to automatically deallocate array components of
the private data, allocated by LMS setup at the end of the solution process.
We use square brackets [ ] to indicate OPTIONAL arguments.
2.2.1 The initialization subroutine
Default values are provided as follows:
CALL LMS initialize( data, control, inform )
data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
problem being solved.
control is a scalar INTENT(OUT) argument of type LMS control type (see §2.1.1). On exit, control contains
default values for the components as described in §2.1.1. These values should only be changed after calling
LMS initialize.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see Section 2.1.3). A successful call to
LMS initialize is indicated when the component status has the value 0. For other return values of status,
see Section 2.3.
2.2.2 The subroutine for setting up the required data structures
The data structures needed to represent the limited-memory matrix Hk or its inverse are set up as follows:
n

CALL LMS setup( n, data, control, inform )
is a scalar INTENT(IN) argument of type default INTEGER, that must be set to the dimension of the limitedmemory matrix required. Restriction: n ≥ 1.

data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
factors obtained. It must not have been altered by the user since the last call to LMS initialize.
control is a scalar INTENT(IN) argument of type LMS control type (see §2.1.1). Default values may be assigned
by calling LMS initialize prior to the first call to LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see §2.1.3). A successful call to LMS setup is
indicated when the component status has the value 0. For other return values of status, see §2.3.
2.2.3 The subroutine for updating the limited memory matrix
The required limited memory matrix is updated to accommodate the incoming triple (sk , yk , δk ) as follows:
S
Y

CALL LMS form( S, Y, delta, data, control, inform[, lambda] )
is an INTENT(IN) rank-1 array of type REAL (double precision in GALAHAD LMS double)and length at least as
large as the value n as set on input to LMS setup, whose first n components must hold the incoming vector sk .
is an INTENT(IN) rank-1 array of type REAL (double precision in GALAHAD LMS double)and length at least as
large as the value n as set on input to LMS setup, whose first n components must hold the incoming vector yk .
Restriction: the update will be skipped for for limited-memory BFGS methods if the inner product sTk yk <= 0.

delta is an INTENT(IN) REAL (double precision in GALAHAD LMS double)scalar that must hold the value δk . Restriction: the update will be skipped if delta ≤ 0.

data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
factors obtained. It must not have been altered by the user since the last call to LMS setup.
control is a scalar INTENT(IN) argument of type LMS control type (see §2.1.1). Default values may be assigned
by calling LMS initialize prior to the first call to LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see §2.1.3). A successful call to LMS setup is
indicated when the component status has the value 0. For other return values of status, see §2.3.
lambda is an OPTIONAL, INTENT(IN) REAL (double precision in GALAHAD LMS double)scalar that if present will be
used to specify the shift λk that is used by the limited memory methods defined by control%method = 1, 2 or
4. Restriction: the update will be skipped if lambda < 0 for these methods.

2.2.4 The subroutine for shifting the limited-memory matrix
The required limited memory matrix is updated to accommodate the shift λk as follows—this call is mandatory when
control%method = 4 if λk was not set during the call to LMS form:
CALL LMS form shift( lambda, data, control, inform )
lambda is an INTENT(IN) REAL (double precision in GALAHAD LMS double)scalar that must hold the value λk . Restriction: the update will be skipped if lambda < 0 or if control%method = 3.
data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
factors obtained. It must not have been altered by the user since the last call to LMS form.
control is a scalar INTENT(IN) argument of type LMS control type (see §2.1.1). Default values may be assigned
by calling LMS initialize prior to the first call to LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see §2.1.3). A successful call to LMS setup is
indicated when the component status has the value 0. For other return values of status, see §2.3.

2.2.5 The subroutine for changing the method defining the limited-memory matrix
The required limited memory matrix is updated to accommodate the shift λk as follows—this call is only permitted if
control%any method = .TRUE. was set when LMS setup was originally called:
CALL LMS change method( data, control, inform, lambda )
data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
factors obtained. It must not have been altered by the user since the last call to LMS form.
control is a scalar INTENT(IN) argument of type LMS control type (see §2.1.1). Default values may be assigned
by calling LMS initialize prior to the first call to LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see §2.1.3). A successful call to LMS setup is
indicated when the component status has the value 0. For other return values of status, see §2.3.
lambda is an OPTIONAL, INTENT(IN) REAL (double precision in GALAHAD LMS double)scalar that if present will be
used to specify the shift λk that is used by the limited memory methods defined by control%method = 1, 2 or
4. Restriction: the update will be skipped if lambda < 0 for these methods.

2.2.6 The subroutine for applying the limited-memory formula to a vector
Given the vector v, the required limited-memory formula, as specified in the most recent call to LMS form, LMS form shift or LMS change method, is applied to v as follows:
CALL LMS apply( V, U, data, control, inform )
V

is a rank-one INTENT(IN) array of type default REAL that must be set on entry to hold the components of the
vector v.

U

is a rank-one INTENT(OUT) array of type default REAL that will be set on exit to the result of applying the
required limited-memory formula to v.

data is a scalar INTENT(INOUT) argument of type LMS data type (see §2.1.4). It is used to hold data about the
factors obtained. It must not have been altered by the user since the last call to LMS setup.
control is a scalar INTENT(IN) argument of type LMS control type (see §2.1.1). Default values may be assigned
by calling LMS initialize prior to the first call to LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type (see §2.1.3). A successful call to LMS apply is
indicated when the component status has the value 0. For other return values of status, see §2.3.
2.2.7 The termination subroutine
All previously allocated arrays are deallocated as follows:
CALL LMS terminate( data, control, inform )
data is a scalar INTENT(INOUT) argument of type LMS data type exactly as for LMS setup, which must not have
been altered by the user since the last call to LMS initialize. On exit, array components will have been
deallocated.
control is a scalar INTENT(IN) argument of type LMS control type exactly as for LMS setup.
inform is a scalar INTENT(OUT) argument of type LMS inform type exactly as for LMS setup. Only the component
status will be set on exit, and a successful call to LMS terminate is indicated when this component status
has the value 0. For other return values of status, see §2.3.
2.3 Warning and error messages
A negative value of inform%status on exit from LMS setup, LMS apply or LMS terminate indicates that an error
has occurred. No further calls should be made until the error has been corrected. Possible values are:
-1. An allocation error occurred. A message indicating the offending array is written on unit control%error, and the
returned allocation status and a string containing the name of the offending array are held in inform%alloc status and inform%bad alloc respectively.
-2. A deallocation error occurred. A message indicating the offending array is written on unit control%error and
the returned allocation status and a string containing the name of the offending array are held in inform%alloc status and inform%bad alloc respectively.
-3. One of the restriction n > 0, delta > 0 or lambda ≥ 0. sT y > 0 has been violated and the update has been
skipped.
-10. The matrix cannot be built with from the current vectors {sk } and {yk } and values δk and λk and the update has
been skipped.

All use is subject to licence. See http://galahad.rl.ac.uk/galahad-www/cou.html .
For any commercial application, a separate license must be signed.
6 LMS (February 20, 2017)

GALAHAD

GALAHAD

LMS

-31. A call to subroutine LMS apply has been made without a prior call to LMS form shift or LMS form with
lambda specified when control%method = 4, or LMS form shift has been called when control%method =
3, or LMS change method has been called after control%any method = .FALSE. was specified when calling
LMS setup.
2.4 Further features
In this section, we describe an alternative means of setting control parameters, that is components of the variable
control of type LMS control type (see §2.1.1), by reading an appropriate data specification file using the subroutine
LMS read specfile. This facility is useful as it allows a user to change LMS control parameters without editing and
recompiling programs that call LMS.
A specification file, or specfile, is a data file containing a number of ”specification commands”. Each command
occurs on a separate line, and comprises a ”keyword”, which is a string (in a close-to-natural language) used to identify
a control parameter, and an (optional) ”value”, which defines the value to be assigned to the given control parameter.
All keywords and values are case insensitive, keywords may be preceded by one or more blanks but values must not
contain blanks, and each value must be separated from its keyword by at least one blank. Values must not contain more
than 30 characters, and each line of the specfile is limited to 80 characters, including the blanks separating keyword
and value.
The portion of the specification file used by LMS read specfile must start with a ”BEGIN LMS” command and
end with an ”END” command. The syntax of the specfile is thus defined as follows:
( .. lines ignored by LMS_read_specfile .. )
BEGIN LMS
keyword
value
.......
.....
keyword
value
END
( .. lines ignored by LMS_read_specfile .. )
where keyword and value are two strings separated by (at least) one blank. The “BEGIN LMS” and “END” delimiter
command lines may contain additional (trailing) strings so long as such strings are separated by one or more blanks,
so that lines such as
BEGIN LMS SPECIFICATION
and
END LMS SPECIFICATION
are acceptable. Furthermore, between the “BEGIN LMS” and “END” delimiters, specification commands may occur in
any order. Blank lines and lines whose first non-blank character is ! or * are ignored. The content of a line after a !
or * character is also ignored (as is the ! or * character itself). This provides an easy manner to ”comment out” some
specification commands, or to comment specific values of certain control parameters.
The value of a control parameters may be of three different types, namely integer, logical or real. Integer and real
values may be expressed in any relevant Fortran integer and floating-point formats (respectively). Permitted values for
logical parameters are ”ON”, ”TRUE”, ”.TRUE.”, ”T”, ”YES”, ”Y”, or ”OFF”, ”NO”, ”N”, ”FALSE”, ”.FALSE.” and ”F”.
Empty values are also allowed for logical control parameters, and are interpreted as ”TRUE”.
The specification file must be open for input when LMS read specfile is called, and the associated device number
passed to the routine in device (see below). Note that the corresponding file is REWINDed, which makes it possible
to combine the specifications for more than one program/routine. For the same reason, the file is not closed by
LMS read specfile.

2.4.1 To read control parameters from a specification file
Control parameters may be read from a file as follows:
CALL LMS_read_specfile( control, device )
control is a scalar INTENT(INOUT)argument of type LMS control type (see §2.1.1). Default values should have
already been set, perhaps by calling LMS initialize. On exit, individual components of control may have
been changed according to the commands found in the specfile. Specfile commands and the component (see
§2.1.1) of control that each affects are given in Table 2.1.
command
error-printout-device
printout-device
print-level
limited-memory-length
limited-memory-method
allow-any-method
space-critical
deallocate-error-fatal
output-line-prefix

component of control
%error
%out
%print level
%memory length
%method
%any method
%space critical
%deallocate error fatal
%prefix

value type
integer
integer
integer
integer
integer
logical
logical
logical
character

Table 2.1: Specfile commands and associated components of control.

device is a scalar INTENT(IN)argument of type default INTEGER, that must be set to the unit number on which the
specfile has been opened. If device is not open, control will not be altered and execution will continue, but
an error message will be printed on unit control%error.
2.5 Information printed
If control%print level is positive, information about the progress of the algorithm will be printed on unit control%out. If control%print level ≥ 1, statistics concerning the formation of R as well as warning and error messages
will be reported.

3 GENERAL INFORMATION
Use of common: None.
Workspace: Provided automatically by the module.
Other routines called directly: None.
Other modules used directly: LMS calls the GALAHAD packages GALAHAD CLOCK, GALAHAD SYMBOLS,
GALAHAD SPACE, GALAHAD LAPACK interface, GALAHAD BLAS interface and GALAHAD SPECFILE,
Input/output: Output is under control of the arguments control%error, control%out and control%print level.
Restrictions: n > 0, delta > 0, lambda ≥ 0.
Portability: ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

All use is subject to licence. See http://galahad.rl.ac.uk/galahad-www/cou.html .
For any commercial application, a separate license must be signed.
8 LMS (February 20, 2017)

GALAHAD

GALAHAD
4

LMS

METHOD

Given a sequence of vectors {sk } and {yk } and scale factors δk , a limited-memory secant approximation Hk is chosen
so that Hmax(k−m,0) = δk I, Hk− j sk− j = yk− j and kHk− j+1 − Hk− j k is “small” for j = min(k − 1, m − 1), . . ., 0. Different
ways of quantifying “small” distinguish different methods, but the crucial observation is that it is possible to construct
Hk quickly from {sk } and {yk } and δk , and to apply it and its inverse to a given vector v. It is also possible to apply
similar formulae to the “shifted” matrix Hk + λk I that occurs in trust-region methods.
The basic methods are those given by
R. H. Byrd, J. Nocedal and R. B. Schnabel (1994) Representations of quasi-Newton matrices and their use in limited
memory methods. Math. Programming, 63(2) 129–156,
with obvious extensions.

5

EXAMPLE OF USE

Suppose that we generate random vectors {sk } and {yk } and scale factors δk , that we build the limited-memory BFGS
−1
matrix Hk and its inverse H−1
k and that we apply Hk and then Hk to a given vector v. Suppose further, that at some
−1
stage, we instead apply the inverse (Hk + λk I) with λk = 0. Then we may use the following code; of course since
−1
we have the identities v = H−1
k (Hk v) and v = (Hk + λk I) (Hk v) when λk = 0, we expect to recover the original v
after every step:
! THIS VERSION: GALAHAD 2.6 - 12/06/2014 AT 15:30 GMT.
PROGRAM GALAHAD_LMS_example
USE GALAHAD_LMS_double
! double precision version
USE GALAHAD_rand_double
IMPLICIT NONE
INTEGER, PARAMETER :: wp = KIND( 1.0D+0 ) ! set precision
INTEGER, PARAMETER :: n = 5, m = 3
TYPE ( LMS_data_type ) :: data, data2
TYPE ( LMS_control_type ) :: control, control2
TYPE ( LMS_inform_type ) :: inform, inform2
REAL ( KIND = wp ), DIMENSION( n ) :: S, Y, U, V
INTEGER :: iter, fail
REAL ( KIND = wp ) :: delta, lambda, sts, stv, ratio, error
TYPE ( RAND_seed ) :: seed
CALL RAND_initialize( seed ) ! Initialize the random generator word
CALL LMS_initialize( data, control, inform ) ! initialize data
control%memory_length = m ! set the memory length
control2 = control
control%method = 1 ! start with L-BFGS
CALL LMS_setup( n, data, control, inform )
control2%method = 3 ! then inverse L-BFGS
control2%any_method =.TRUE. ! allow the 2nd update to change method
CALL LMS_setup( n, data2, control2, inform2 )
fail = 0 ! count the failures
DO iter = 1, 5 * n
IF ( iter == 3 * n ) THEN ! switch to inverse shifted L-BFGS
CALL LMS_setup( n, data, control, inform )
control2%method = 4
CALL LMS_setup( n, data2, control2, inform2 )
END IF
CALL RAND_random_real( seed, .FALSE., S ) ! pick random S, Y and delta
CALL RAND_random_real( seed, .FALSE., Y )

IF ( DOT_PRODUCT( S, Y ) < 0.0_wp ) Y = - Y ! ensure that SˆT Y is positive
CALL RAND_random_real( seed, .TRUE., delta )
CALL LMS_form( S, Y, delta, data, control, inform ) ! update the model
IF ( inform%status /= 0 ) THEN
WRITE( 6, "( ’ update error, status = ’, I0 )" ) inform%status
fail = fail + 1 ; CYCLE
END IF
V = 1.0_wp ! form the first product with the vector ones
CALL LMS_apply( V, U, data, control, inform ) ! form the required product
IF ( inform%status /= 0 ) THEN
WRITE( 6, "( ’ apply error, status = ’, I0 )" ) inform%status
fail = fail + 1 ; CYCLE
END IF
CALL LMS_form( S, Y, delta, data2, control2, inform2 ) ! update model 2
IF ( inform2%status /= 0 ) THEN
WRITE( 6, "( ’ update error, status = ’, I0 )" ) inform2%status
fail = fail + 1 ; CYCLE
END IF
IF ( control2%method == 4 ) THEN
lambda = 0.0_wp ! apply the shifted L_BFGS (inverse) with zero shift
CALL LMS_form_shift( lambda, data2, control2, inform2 )
IF ( inform2%status /= 0 ) THEN
WRITE( 6, "( ’ update error, status = ’, I0 )" ) inform2%status
fail = fail + 1 ; CYCLE
END IF
END IF
! note, the preceeding two calls could have been condensed as
!
CALL LMS_form( S, Y, delta, data2, control2, inform2, lambda = 0.0_wp )
CALL LMS_apply( U, V, data2, control2, inform2 ) ! form the new product
IF ( inform2%status /= 0 ) THEN
WRITE( 6, "( ’ apply error, status = ’, I0 )" ) inform2%status
fail = fail + 1 ; CYCLE
END IF
IF ( MAXVAL( ABS( V - 1.0_wp ) ) > 0.00001_wp ) fail = fail + 1
END DO
IF ( fail == 0 ) THEN ! check for overall success
WRITE( 6, "( ’ no failures ’ )" )
ELSE
WRITE( 6, "( I0, ’ failures ’ )" ) fail
END IF
CALL LMS_terminate( data, control, inform ) ! delete internal workspace
CALL LMS_terminate( data2, control2, inform2 )
END PROGRAM GALAHAD_LMS_example

This produces the following output:
no failures

